*******************************************************************************
*The SAS program (HEI2010_NHANES0708_MC_PopulationScore.sas):	              *
*									      *
*This program creates component and total scores of the HEI-2010	      * 
*for a population or a group.						      *
*The 12 components include: Total Fruit, Whole Fruit, 	      	              *	
*Total Vegetables, Greens and Beans, Whole Grains,  			      *	
*Dairy, Total Protein Foods, Seafood and Plant Proteins, Fatty Acids,	      *
*Refined Grains, Sodium, and Empty Calories.				      *
*									      *
*Please see accompanying readme file.   				      *
*******************************************************************************;
  
/* Libname and filename for the input datasets.*/

libname demo xport 'C:/DEMO_E.xpt'; /*this is the location of the NHANES DEMO_E SAS transport file*/
libname tot xport 'C:/DR1TOT_E.xpt'; /*this is the location of the NHANES DR1TOT_E SAS transport file*/
libname iff xport 'C:/DR1IFF_E.xpt'; /*this is the location of the NHANES DR1IFF_E SAS transport file*/
libname mped 'C:/mped'; /*this is the location of the MPED SAS dataset and the CNPP MyPyramid Equivalents Database for Whole Fruit and Fruit Juice SAS dataset*/
filename cnppadd 'C:/CNPP-MPED2point0B-Addendum.xls'; /*this is the location of the CNPP Addendum to the MyPyramid Equivalents Database (MPED), version 2.0B */

/*Name and location of the output file to be exported, containing mean HEI-2010 component and total scores and their confidence intervals*/

filename res 'C:/hei2010.result.csv'; 

/*Read in required macros*/

%include 'C:/hei2010.beanspeas.allocation.macro.sas';
%include 'C:/hei2010.score.macro.sas';
      
TITLE 'HEI-2010 scores for NHANES 2007-2008, AGE >= 2, RELIABLE DIETS, Include Pregnant and Lactating Women';

/*Section (I): Calculations at the individual participant level to obtain variables needed to calculate HEI-2010 scores.*/

/*Step 1: locate the required datasets and variables and make necessary exclusions*/

*get mped version 2.0;
data mped;
 set mped.MyPyrEquivDB_v2;
 foodcode=1*drdifdcd;
 modcode=1*drdimc;

 *delete 5 foodcodes and use values from the Addendum;
 if foodcode in (11710800, 11710801, 58106210, 58106220, 58106230) then delete; 
run;

proc sort data=mped;
  by foodcode modcode;
run;

*get cnpp fruit info for mped 2.0;
data cnpp;
  set mped.cnppmyp_v1nhanes0304_wjfrt;
run;

proc sort data=cnpp;
  by foodcode modcode;
run;
  
*get cnpp addendum version 2.0 B;
proc import datafile=cnppadd
  out=cnppadd
  dbms=xls
  replace;
  getnames=yes;
run;

proc sort data=cnppadd;
  by foodcode modcode;
run;

data newmped;
  merge cnppadd mped cnpp;
  by foodcode modcode;
run;
  
*move soy;
data newmped;
  set newmped;

  /*In the HEI-2010, soy beverages are counted as part of the Milk component.  
 Convert the four soy beverage codes in the MPED from M_SOY oz equivalents to D_TOTAL cup 
 equivalents using the following conversion process*/

  /*FOODCODE=11310000, MILK, IMITATION, FLUID, SOY BASED (1 cup=244 grams)
    FOODCODE=11320000, MILK, SOY, READY-TO-DRINK, NOT BABY (1 cup=245 grams)
    FOODCODE=11321000, MILK, SOY, READY-TO-DRINK, NOT BABY'S, CHOCOLATE (1 cup=240 grams)
    FOODCODE=11330000, MILK, SOY, DRY, RECONSTITUTED, NOT BABY (1 cup=245 grams)*/

  IF FOODCODE=11310000 THEN DO;
     M_SOY=0;
	 D_TOTAL=ROUND(100*(1/244),.001);
	 END;
  ELSE IF FOODCODE=11320000 THEN DO;
     M_SOY=0;
	 D_TOTAL=ROUND(100*(1/245),.001);
	 END;
  ELSE IF FOODCODE=11321000 THEN DO;
     M_SOY=0;
	 D_TOTAL=ROUND(100*(1/240),.001);
	 END;
  ELSE IF FOODCODE=11330000 THEN DO;
     M_SOY=0;
	 D_TOTAL=ROUND(100*(1/245),.001);
	 END;
run;

PROC SORT DATA=newmped nodup;
  BY FOODCODE modcode;
run;

DATA Food;                           
  SET IFF.DR1IFF_E;
  FOODCODE=1*DR1IFDCD; /*convert variable name and type*/
  if DR1DRSTZ=1; /*reliable dietary recall status*/
  modcode=1*dr1mc;
run;

PROC SORT DATA=Food;
  BY FOODCODE modcode;
run;

DATA Nutrient (keep=seqn WTDRD1 dr1TKCAL dr1TCARB dr1TSFAT dr1TALCO DR1TSODI DR1DRSTZ
    dr1tmfat dr1tpfat dr1tsfat);
  SET tot.DR1TOT_E;
run;

PROC SORT DATA=Nutrient;
  BY SEQN;
run;

DATA demo (keep=seqn RIDAGEYR RIAGENDR SDDSRVYR SDMVPSU SDMVSTRA);
  SET DEMO.DEMO_E;
run;

PROC SORT DATA=demo;
  BY SEQN;
run;


/*Step 2: Combine the required datasets and make necessary adjustments at the level of individual foods. */

data cohort;
  merge Nutrient demo;
  by seqn;
run;

data cohort;
  set cohort;
  if DR1DRSTZ=1; /*reliable dietary recall status*/
  if ridageyr >= 2;
run;

DATA FDPYR;
  MERGE Food (IN=S) NEWMPED (IN=N);
  BY FOODCODE modcode;
  IF S AND N;
run;

/*Convert NHANES 07-08 individuals' food intake amounts from grams to 
MyPyramid equivalents*/

DATA FDPYR;
  SET FDPYR;

/*Calculate intake for MyPyramid food groups*/  

 ARRAY PYRVAR G_TOTAL--A_BEV WHOLEFRT FRTJUICE;

 DO OVER PYRVAR;

    PYRVAR=PYRVAR*(dr1IGRMS/100);
 END;

run;

PROC SORT DATA=FDPYR;
  BY SEQN;
run;

/*Calculate individual's food intake amounts for MyPyramid food groups in 1 day*/
PROC MEANS DATA=FDPYR NOPRINT;
  BY SEQN;
  VAR G_TOTAL--A_BEV WHOLEFRT FRTJUICE;
  OUTPUT OUT=PYRCALC SUM= ;
run;

/*Combine all required datasets*/

DATA BOTH;
  MERGE cohort (in=c) Nutrient (IN=F) PYRCALC(IN=P) DEMO;
  BY SEQN;
  IF c;   
run;

/*Please note: MyPyramid equivalent values for total vegetable intake (V_TOTAL) in the HEI-2010
  may be different from V_TOTAL in the MPED because legumes may be counted as vegetables or meat
  in the HEI-2010; and total dairy intake (D_TOTAL) in the HEI-2010 may be different from D_TOTAL 
  in the MPED because soy beverages are counted as milk in the HEI-2010.*/

DATA BOTH;
  SET BOTH;
  BY SEQN;

  *set to zero for people with zero cal who do not appear in iff data;
 ARRAY PYRVAR G_TOTAL--A_BEV WHOLEFRT FRTJUICE;

 DO OVER PYRVAR;
    if pyrvar < 0 then pyrvar=0;
 END;
run;


/*Step 3: Calculate NHANES 07-08 individual food and nutrient intakes for each individual. */


DATA BOTH;
  SET BOTH;
  BY SEQN;

  MONOPOLY=dr1tmfat+dr1tpfat;

  ALLMEAT=M_MPF+M_EGG+M_NUTSD+M_SOY;
  SEAPLANT=M_FISH_HI+M_FISH_LO+M_SOY+M_NUTSD;

 /**Calculate intake of Empty Calories**/

      ADDSUGC=16*ADD_SUG;	/*calories from added sugars*/
      SOLFATC=DISCFAT_SOL*9;	/*calories from solid fat*/

      maxalcgr=13*(dr1tkcal/1000);  /*max grams of alcohol based on kcal intake*/

      *ETHCAL=DR1IALCO*7;
      if dr1TALCO <= maxalcgr then EXALCCAL=0;  /*consumed less than max*/
        else if dr1TALCO > maxalcgr then EXALCCAL=7*(dr1TALCO-maxalcgr);  /*get cal from extra alc grams*/

      EMPTYCAL10=ADDSUGC+SOLFATC+EXALCCAL; /*total empty calories in hei2010 definition*/
                           
run; 

/*Section (II): Calculation of weighted means and a variance-covariance matrix and generation
 of a Monte Carlo dataset, enabling standard errors to be calculated.*/


/*Step 1.  Calculate the weighted means and the variance/covariance matrix for the dietary variables of interest. */
data one;
    set both;
    array comp (15) dr1TKCAL v_total v_drkgr legumes f_total wholefrt g_whl d_total 
      allmeat seaplant monopoly dr1tsfat DR1TSODI G_NWHL EMPTYCAL10;
    * Turn each variable into an observation of the single variable VBL ;
    * keeping track of the order with the dum_num variable; 
    do i = 1 to 15;
        VBL = comp(i);
        dum_num = i;
        output;
        end;
run;

data one;
    set one;
    * Create dummies with the same names as the original variables;
    * The i-th dummy gets a 1 if the observation is associated with the
    * i-th original variable;
    array comp (15) dr1TKCAL v_total v_drkgr legumes f_total wholefrt g_whl d_total 
      allmeat seaplant monopoly dr1tsfat DR1TSODI G_NWHL EMPTYCAL10;
    do i = 1 to 15;
        if dum_num = i then comp(i) = 1;
        else comp(i) = 0;
        end;
    drop i dum_num;
run;
* Now run PROC (SURVEY)REG with dependent variable VBL and all the dummies;
* as predictors. Force the regression to not include an intercept, and ask;
* for the covariance matrix of the parameter estimates. ;

proc surveyreg data=one;
    strata SDMVSTRA;
    cluster SDMVPSU;
    weight WTDRD1;
    model VBL=dr1TKCAL v_total v_drkgr legumes f_total wholefrt g_whl d_total 
      allmeat seaplant monopoly dr1tsfat DR1TSODI G_NWHL EMPTYCAL10/noint covb;
    * Output the covariance matrix we wanted all along;
    ods output covb=csd_cov;
    title2 "Tricking SURVEYREG into giving us the covariance matrix of means";
run;

proc print data=csd_cov;
    title2 "Printout of csd_cov dataset -uses complex survey info";
run;

proc means data=both n min max mean;
  weight wtdrd1;
  var dr1TKCAL v_total v_drkgr legumes f_total wholefrt g_whl d_total 
      allmeat seaplant monopoly dr1tsfat DR1TSODI G_NWHL EMPTYCAL10;
  title2 'look at weighted means';
  output out=wtdm mean= ;
run;

data covdata (drop=Parameter);
  set csd_cov;
  _TYPE_='COV ';
  _NAME_=Parameter;
run;

data wtdm (drop=_TYPE_ _FREQ_);
  set wtdm;
run;

data wtdm;
  set wtdm;
  _TYPE_='MEAN';
run;

data covdata;
  set covdata wtdm;
run;

proc print data=covdata;
  title2 'input to simnorml';
run;

/*Step 2.  In this step, a Monte Carlo data set with 10,000 rows is generated using the means
 and variance/covariance matrix from step 1*/

proc simnormal data=covdata(type=cov) numreal=10000 seed=51230077 outseed out=sim_data;
  var dr1TKCAL v_total v_drkgr legumes f_total wholefrt g_whl d_total 
      allmeat seaplant monopoly dr1tsfat DR1TSODI G_NWHL EMPTYCAL10;
run;

proc means data=sim_data n nmiss min max mean stddev;
  var dr1TKCAL v_total v_drkgr legumes f_total wholefrt g_whl d_total 
      allmeat seaplant monopoly dr1tsfat DR1TSODI G_NWHL EMPTYCAL10;
  title2 "Distributions of Simulated Data";
run;

proc print data=sim_data(obs=20);
  title2 "Listing of 20 Records from Simulated Data";
run;

/*Section (III): Allocation of beans and peas (legumes) and application of the HEI-2010 scoring algorithm.*/

/*Step 1. Allocate beans and peas (legumes) using the Monte Carlo data set from Section II and the beans and peas allocation macro. */

%LEG2010A (indat=sim_data,
           kcal=dr1TKCAL,
           allmeat=allmeat,
	   seaplant=seaplant,
	   v_total=v_total,
	   v_drkgr=v_drkgr,
	   legumes=legumes,
	   outdat=lsim_data);


/*Step 2. Apply the HEI-2010 scoring macro. */

%HEI2010 (indat=lsim_data,
          kcal=dr1TKCAL,
	  lv_total=legume_added_V_TOTAL,
	  lbeangrn=legume_added_BEANGRN,
	  f_total=f_total,
	  wholefrt=wholefrt,
	  g_whl=g_whl,
	  d_total=d_total,
          lallmeat=legume_added_ALLMEAT,
	  lseaplant=legume_added_SEAPLANT,
	  monopoly=monopoly,
	  sfat=dr1tsfat,
	  sodi=DR1TSODI,
	  G_NWHL=G_NWHL,
	  EMPTYCAL10=EMPTYCAL10,
	  outdat=aftermac);

run;

proc means data=aftermac n nmiss min max mean stddev;
  var legume_added_v_total legume_added_BEANGRN legume_added_allmeat legume_added_SEAPLANT 
    vegden grbnden frtden whfrden wgrnden dairyden
     meatden seaplden faratio sodden rgden sofa_perc;
  title2 'after legume allocation and hei 2010 scoring macro';
run;

/*Section (IV): Calculation of HEI-2010 component and total scores and their confidence intervals.*/

/*Step 1.  This step uses univariate and means procedures to compute total and component scores and their standard errors.  */
proc univariate data=aftermac noprint;
  var HEIX1_TOTALVEG HEIX2_GREEN_AND_BEAN HEIX3_TOTALFRUIT HEIX4_WHOLEFRUIT HEIX5_WHOLEGRAIN HEIX6_TOTALDAIRY 
    HEIX7_TOTPROT HEIX8_SEAPLANT_PROT HEIX9_FATTYACID HEIX10_SODIUM HEIX11_REFINEDGRAIN HEIX12_SOFAAS HEI2010_TOTAL_SCORE;
  output out=ci pctlpts=2.5 97.5 pctlpre=h1_ h2_ h3_ h4_ h5_ h6_ h7_ h8_ h9_ h10_ h11_ h12_ totscore_;
run;

proc means data=aftermac noprint;
  var HEIX1_TOTALVEG HEIX2_GREEN_AND_BEAN HEIX3_TOTALFRUIT HEIX4_WHOLEFRUIT HEIX5_WHOLEGRAIN HEIX6_TOTALDAIRY 
    HEIX7_TOTPROT HEIX8_SEAPLANT_PROT HEIX9_FATTYACID HEIX10_SODIUM HEIX11_REFINEDGRAIN HEIX12_SOFAAS HEI2010_TOTAL_SCORE;
  output out=stat min=h1_min h2_min h3_min h4_min h5_min h6_min h7_min h8_min h9_min h10_min h11_min h12_min totscore_min
   max=h1_max h2_max h3_max h4_max h5_max h6_max h7_max h8_max h9_max h10_max h11_max h12_max totscore_max
    mean=h1_mean h2_mean h3_mean h4_mean h5_mean h6_mean h7_mean h8_mean h9_mean h10_mean h11_mean h12_mean totscore_mean
     stddev= h1_stddev h2_stddev h3_stddev h4_stddev h5_stddev h6_stddev h7_stddev h8_stddev h9_stddev h10_stddev h11_stddev h12_stddev totscore_stddev;  
run;

/*Step 2.  This step prepares the results for display*/
data all;
  merge ci stat;
run;

data result (keep=score slabel min max mean stderr low high);
  set all;
  score='HEIx1          ';
  slabel='HEI-2010 COMPONENT 1 TOTAL VEGETABLES';
  min=h1_min;
  max=h1_max;
  mean=h1_mean;
  stderr=h1_stddev;
  low=h1_2_5;
  high=h1_97_5;
  output result;
  score='HEIx2';
  slabel='HEI-2010 COMPONENT 2 GREENS AND BEANS';
  min=h2_min;
  max=h2_max;
  mean=h2_mean;
  stderr=h2_stddev;
  low=h2_2_5;
  high=h2_97_5;
  output result;
  score='HEIx3';
  slabel='HEI-2010 COMPONENT 3 TOTAL FRUIT';
  min=h3_min;
  max=h3_max;
  mean=h3_mean;
  stderr=h3_stddev;
  low=h3_2_5;
  high=h3_97_5;
  output result;
  score='HEIx4';
  slabel='HEI-2010 COMPONENT 4 WHOLE FRUIT';
  min=h4_min;
  max=h4_max;
  mean=h4_mean;
  stderr=h4_stddev;
  low=h4_2_5;
  high=h4_97_5;
  output result;
  score='HEIx5';
  slabel='HEI-2010 COMPONENT 5 WHOLE GRAINS';
  min=h5_min;
  max=h5_max;
  mean=h5_mean;
  stderr=h5_stddev;
  low=h5_2_5;
  high=h5_97_5;
  output result;
  score='HEIx6';
  slabel='HEI-2010 COMPONENT 6 DAIRY';
  min=h6_min;
  max=h6_max;
  mean=h6_mean;
  stderr=h6_stddev;
  low=h6_2_5;
  high=h6_97_5;
  output result;
  score='HEIx7';
  slabel='HEI-2010 COMPONENT 7 TOTAL PROTEIN FOODS';
  min=h7_min;
  max=h7_max;
  mean=h7_mean;
  stderr=h7_stddev;
  low=h7_2_5;
  high=h7_97_5;
  output result;
  score='HEIx8';
  slabel='HEI-2010 COMPONENT 8 SEAFOOD AND PLANT PROTEIN';
  min=h8_min;
  max=h8_max;
  mean=h8_mean;
  stderr=h8_stddev;
  low=h8_2_5;
  high=h8_97_5;
  output result;
  score='HEIx9';
  slabel='HEI-2010 COMPONENT 9 FATTY ACID RATIO';
  min=h9_min;
  max=h9_max;
  mean=h9_mean;
  stderr=h9_stddev;
  low=h9_2_5;
  high=h9_97_5;
  output result;
  score='HEIx10';
  slabel='HEI-2010 COMPONENT 10 SODIUM';
  min=h10_min;
  max=h10_max;
  mean=h10_mean;
  stderr=h10_stddev;
  low=h10_2_5;
  high=h10_97_5;
  output result;
  score='HEIx11';
  slabel='HEI-2010 COMPONENT 11 REFINED GRAINS';
  min=h11_min;
  max=h11_max;
  mean=h11_mean;
  stderr=h11_stddev;
  low=h11_2_5;
  high=h11_97_5;
  output result;
  score='HEIx12';
  slabel='HEI-2010 COMPONENT 12 SOFAAS CALORIES';
  min=h12_min;
  max=h12_max;
  mean=h12_mean;
  stderr=h12_stddev;
  low=h12_2_5;
  high=h12_97_5;
  output result;
  score='TOTAL HEI 2010';
  slabel='TOTAL HEI-2010 SCORE';
  min=totscore_min;
  max=totscore_max;
  mean=totscore_mean;
  stderr=totscore_stddev;
  low=totscore_2_5;
  high=totscore_97_5;
  output result;
run;

/*Step 3 - This step displays the results*/
proc print data=result;
  id score;
  var slabel min max mean stderr low high;
  title2 'complex survey design population method - mean and confidence interval of HEI-2010 using NH 07/08 data';
run;

/*Step 4 - This step provides an option to export the results into a CSV file that can be opened in Excel.*/

proc export data=result
  file=res
  dbms=csv
  replace;
run;




